
PrintForm.NET

By Rod Stephens

If you turn your back on Rod Stephens for a minute, he starts writing graphics code. This month he turns to printing and retools the old VB 6 favorite PrintForm for VB .NET. He then shows how to build a generic high-resolution form printing routine that makes printing high-quality images of forms a snap.


VB 6’s PrintForm method lets you print a form in a single statement. The result is somewhat grainy because it’s a pixel-by-pixel dump of the form’s image but it’s fantastically simple. If your deadline is looming and you’re frantically slapping together a data entry form with a hundred or so labels and text boxes, “simple” is the word you want to hear.


Sadly, VB .NET abandoned PrintForm during its purge of all things VB-specific. To print a form in .NET, you need to write a ton of code to explicitly position every label, text box, and radio button you want drawn by the printer. Or do you? While VB .NET doesn’t give you a nice simple PrintForm method, it does give you the tools you need to build your own.

PrintForm.NET


I haven’t yet found a way to make VB .NET itself capture a form’s image. It’s not hard using API functions, however. Once you have the form’s image, you can do whatever you want with it including displaying it as a print preview or printing it yourself. Program NetPrintForm, available for download in the subscriber’s area, uses a PrintPreviewDialog control to display a form’s image. The control lets you zoom in, scroll, or print the result.


Listing 1 shows the program’s most interesting code. The GetWindowImage function takes as a parameter a window handle. It uses the GetWindowRect API function to get the window’s dimensions and makes a new Bitmap object to fit. It uses GetWindowDC to get the window’s device context (DC) and then calls BitBlt to copy the window’s image into the Bitmap. At that point, it’s done with API calls and can work in VB .NET code again.


Subroutine PrintPreviewForm uses GetWindowImage to fetch an image of the form. It then makes a new PrintDocument object and saves it in the variable m_PrintDocument. This variable is declared using the WithEvents keyword so the program can capture its events. The routine sets the PrintPreviewDialog control’s Document property to this object and then calls the control’s ShowDialog method to display the print preview.


When it needs to draw a page, the PrintPreviewDialog control raises the PrintDocument object’s PrintPage event. The event handler converts the form’s size from pixels into hundredths of an inch, the units used by the printer. Depending on the its enlarge_to_fit parameter, the routine either centers the form’s image on the page at its normal size or enlarges it as much as possible without changing its aspect ratio. The dialog takes care of the rest.

' The form's image.

Private m_Bitmap As Bitmap

Private m_EnlargeToFit As Boolean

' The PrintDocument object that does the printing.

Private WithEvents m_PrintDocument As PrintDocument

' Return a Bitmap containing the window's image.

Public Function GetWindowImage(ByVal hWnd As Integer) As Bitmap

    ' Get the window's rectangle.

    Dim r As New RECT

    GetWindowRect(hWnd, r)

    Dim wid As Integer = r.Right - r.Left

    Dim hgt As Integer = r.Bottom - r.Top

    ' Make a Bitmap of the right size and 

    ' an associated Graphics object.

    Dim bm As New Bitmap(wid, hgt)

    Dim gr As Graphics = Graphics.FromImage(bm)

    ' Copy the image from the window to the Bitmap.

    Dim src_hdc As Integer = GetWindowDC(hWnd)

    Dim bm_hdc As IntPtr = gr.GetHdc

    Const SRCCOPY As Integer = &HCC0020

    BitBlt(bm_hdc.ToInt32, 0, 0, wid, hgt, _

        src_hdc, 0, 0, SRCCOPY)

    ' Clean up.

    gr.ReleaseHdc(bm_hdc)

    ReleaseDC(hWnd, src_hdc)

    ' Return the bitmap.

    Return bm

End Function

' Display a print preview of the form.

Public Sub PrintPreviewForm(ByVal dlg As PrintPreviewDialog, _

 ByVal frm As Form, Optional ByVal enlarge_to_fit As Boolean = False)

    ' Save the form's image now so the PrintPreview

    ' dialog doesn't obscure part of it.

    m_Bitmap = GetWindowImage(frm.Handle.ToInt32)

    m_EnlargeToFit = enlarge_to_fit

    ' Make the PrintDocument object to print.

    m_PrintDocument = New PrintDocument

    dlg.Document = m_PrintDocument

    dlg.ShowDialog()

End Sub

' Print the form's image.

Private Sub m_PrintDocument_PrintPage(ByVal sender As Object, _

 ByVal e As System.Drawing.Printing.PrintPageEventArgs) _

 Handles m_PrintDocument.PrintPage

    ' Get the image's size in 100ths of inches

    ' (the printer's units).

    Dim wid As Single = m_Bitmap.Width / m_Bitmap.HorizontalResolution * 100

    Dim hgt As Single = m_Bitmap.Height / m_Bitmap.VerticalResolution * 100

    ' Draw the image on the page.

    If Not m_EnlargeToFit Then

        ' Center the image.

        Dim x As Single = (e.MarginBounds.Width - wid) / 2 + e.MarginBounds.Left

        Dim y As Single = (e.MarginBounds.Height - hgt) / 2 + e.MarginBounds.Top

        e.Graphics().DrawImage(m_Bitmap, x, y)

    Else

        ' Enlarge the image as much as possible

        ' but preserve the aspect ratio.

        Dim dest_rect As RectangleF

        If wid / hgt > e.MarginBounds.Width / e.MarginBounds.Height Then

            ' The image is relatively short and wide.

            ' Make it as wide as possible.

            With dest_rect

                .Width = e.MarginBounds.Width

                .Height = hgt * .Width / wid

                .X = e.MarginBounds.Left

                .Y = e.MarginBounds.Top + (e.MarginBounds.Height - .Height) / 2

            End With

        Else

            ' The image is relatively tall and thin.

            ' Make it as tall as possible.

            With dest_rect

                .Height = e.MarginBounds.Height

                .Width = wid * .Height / hgt

                .X = e.MarginBounds.Left + (e.MarginBounds.Width - .Width) / 2

                .Y = e.MarginBounds.Top

            End With

        End If

        ' Draw the image.

        e.Graphics().DrawImage(m_Bitmap, dest_rect)

    End If

    ' Debugging: Draw the margins.

    'e.Graphics().DrawRectangle(Pens.Red, _

    '    e.MarginBounds.Left, e.MarginBounds.Top, _

    '    e.MarginBounds.Width, e.MarginBounds.Height)

    ' Discard the image.

    m_Bitmap = Nothing

    ' We don't need any more PrintDocument events.

    m_PrintDocument = Nothing

End Sub

Listing 1. Program NetPrintForm displays a form’s image in a print preview.

High-Resolution PrintForm


The PrintPreviewForm subroutine makes printing a form in VB .NET about as easy as using PrintForm in VB 6. It still produces a grainy pixel-by-pixel image of the form, however. If you want a really high-quality printout, you can’t just grab an image of the form. Instead you need to draw each of the form’s controls yourself exactly where you want them using the fonts, styles, and colors you want them to have.


To take full advantage of the printer’s surface, you often need to make adjustments to the form’s layout. For example, a form may hold a list or grid control that isn’t big enough to display all of its values on the form at the same time. Because a piece of paper is generally taller than a form on the screen, you may be able to give those controls the extra space they deserve and display all of the data at once. For example, an order form might have a series of text fields at the top and then a long list of order items filling the page below.


For a complicated form, arranging the controls so they display as much data as possible while looking somewhat like the on-screen form can take many hours. However, if you just want to print something with high resolution that looks more or less like the form, you can automate much of the work.


The idea is relatively straightforward. The program loops through the controls on the form, drawing each in the appropriate position on the printout.


The form’s Controls collection holds controls in their stacking order so those at the beginning of the collection are at the top of the stacking order. The loops through this collection from back to front so it draws the controls on the bottom of the stack first. That makes the later controls draw over the earlier ones.


Example program NetHiResPrint, also available for download, demonstrates this method to make a a high-resolution printout. Listing 2 shows the program’s most interesting pieces of code.


Figuring out how to map each control’s position in pixels to the printed output’s hundredths of inches while centering the form would be a major headache so the code takes a different approach. Subroutine HiResPrint starts by deciding how it would need to scale and translate the form to map its pixel positions to the correctly centered position on the printed page. It then uses the ScaleTransform and TranslateTransform methods to define that transformation for the page’s Graphics object. Now the program can use the form’s native pixel units and the Graphics object automatically translates them to the correct positions on the printout.


After it sets up the transformation, subroutine HiResPrint loops through the controls in back-to-front order, calling the DrawControl subroutine for each. HiResPrint finishes by drawing a box around the form if desired.


Subroutine DrawControl takes as parameters the control’s X and Y coordinate offsets x_off and y_off. This is useful for controls contained inside a GroupBox. When a control is contained inside a GroupBox, its X and Y coordinates are relative to the GroupBox not the form. The x_off and y_off parameters give subroutine DrawControl the GroupBox’s position.


DrawControl starts by verifying that the control is visible (no point drawing it otherwise). It calculates the control’s absolute position relative to the form and uses a Select Case statement to call a drawing routine that is appropriate for this kind of control. For example, it calls DrawTextBox to draw TextBoxes.


After it finishes drawing the control, the code checks the control’s Controls collection. If the collection is not empty, as is the case when a GroupBox contains other controls, DrawControl recursively calls itself to draw the contained controls.


The routines that draw the different kinds of controls (TextBox, Label, PictureBox, and so on) vary in detail but DrawTextualControl provides a good example. It calls subroutine DrawBorder to erase the control’s area and draw a border around it if appropriate. It then calls subroutine DrawText to display the control’s text aligned appropriately.


DrawBorder erases the control’s area so the control will cover any controls that were drawn earlier in the stacking order. It then uses reflection to see if the control has a BorderStyle property. If it does, and if BorderStyle is not None, the routine draws a border around the control. DrawBorder also takes a parameter that lets the calling routine force it to draw a border. The code uses this to ensure that Buttons get borders even though they don’t have a BorderStyle property.


Subroutine DrawText draws text inside a control’s area. It starts by creating a layout rectangle that defines the area in which the text should appear. DrawText makes a StringFormat object to indicate how the text should be aligned (left, right, middle, top, bottom, and so forth) and then calls DrawString to draw the text inside the layout rectangle with the proper alignment and word wrapping. (This step is a lot easier in VB .NET than in VB 6 where you would have to figure out how to align and wrap each word in the text yourself.)


The other control drawing routines are similar. They all draw controls in the appropriate position on the form. Their differences come from the different appearances of different kinds of controls.

' Produce a high resolution printout of the form.

Public Sub HiResPrint(ByVal frm As Form, ByVal e As System.Drawing.Printing.PrintPageEventArgs, _

  Optional ByVal enlarge_to_fit As Boolean = False, _

  Optional ByVal draw_box As Boolean = True)

    '' Debugging: Draw the margins in printer units.

    'e.Graphics.DrawRectangle(New Pen(Color.Blue, 5), _

    '    e.MarginBounds.Left, _

    '    e.MarginBounds.Top, _

    '    e.MarginBounds.Width, _

    '    e.MarginBounds.Height)

    Dim scale As Single

    Dim x As Single

    Dim y As Single

    ' Set an appropriate scale transformation.

    If Not enlarge_to_fit Then

        ' Print at normal scale.

        ' Get the form client area's size 

        ' in 100ths of inches (the printer's units).

        Dim bm As New Bitmap(1, 1)

        scale = 100 / bm.HorizontalResolution

        bm = Nothing

        Dim wid As Single = frm.ClientSize.Width * scale

        Dim hgt As Single = frm.ClientSize.Height * scale

        ' Calculate the upper left corner

        ' in printer units.

        x = e.MarginBounds.Left + (e.MarginBounds.Width - wid) / 2

        y = e.MarginBounds.Top + (e.MarginBounds.Height - hgt) / 2

    Else

        ' Enlarge the image as much as possible

        ' but preserve the aspect ratio.

        Dim wid As Integer = frm.ClientSize.Width

        Dim hgt As Integer = frm.ClientSize.Height

        If wid / hgt > e.MarginBounds.Width / e.MarginBounds.Height Then

            ' The image is relatively short and wide.

            ' Make it as wide as possible.

            scale = CSng(e.MarginBounds.Width / wid)

        Else

            ' The image is relatively tall and thin.

            ' Make it as tall as possible.

            scale = CSng(e.MarginBounds.Height / hgt)

        End If

        wid = CInt(wid * scale)

        hgt = CInt(hgt * scale)

        ' Calculate the upper left corner 

        ' in printer units.

        x = e.MarginBounds.Left + (e.MarginBounds.Width - wid) \ 2

        y = e.MarginBounds.Top + (e.MarginBounds.Height - hgt) \ 2

    End If

    ' Set the transformation.

    e.Graphics.ScaleTransform(scale, scale, Drawing2D.MatrixOrder.Append)

    e.Graphics.TranslateTransform(x, y, Drawing2D.MatrixOrder.Append)

    '' Debugging: Draw the margins again in form units.

    'e.Graphics.DrawRectangle(Pens.Red, _

    '    (e.MarginBounds.Left - x) / scale, _

    '    (e.MarginBounds.Top - y) / scale, _

    '    e.MarginBounds.Width / scale, _

    '    e.MarginBounds.Height / scale)

    ' Draw controls in back to front order.

    For i As Integer = frm.Controls.Count - 1 To 0 Step -1

        DrawControl(frm.Controls(i), e.Graphics, 0, 0)

    Next i

    ' Draw a box around the form if desired.

    If draw_box Then

        e.Graphics.DrawRectangle(Pens.Black, 0, 0, _

            frm.ClientSize.Width, frm.ClientSize.Height)

    End If

End Sub

' Draw this control.

Private Sub DrawControl(ByVal ctl As Control, ByVal gr As Graphics, _

 ByVal x_off As Integer, ByVal y_off As Integer)

    ' Don't bother if the control is not visible.

    If Not ctl.Visible Then Exit Sub

    Dim l As Integer = x_off + ctl.Left

    Dim t As Integer = y_off + ctl.Top

    Dim w As Integer = ctl.Width

    Dim h As Integer = ctl.Height

    Select Case TypeName(ctl)

        Case "Button"

            DrawTextualControl(ctl, gr, l, t, w, h, CType(ctl, Button).TextAlign, True)

        Case "Label"

            DrawTextualControl(ctl, gr, l, t, w, h, CType(ctl, Label).TextAlign)

        Case "LinkLabel"

            DrawTextualControl(ctl, gr, l, t, w, h, CType(ctl, LinkLabel).TextAlign)

        Case "TextBox"

            DrawTextBox(ctl, gr, l, t, w, h)

        Case "PictureBox"

            DrawPictureBox(ctl, gr, l, t, w, h)

        Case "CheckBox"

            DrawCheckBox(ctl, gr, l, t, w, h)

        Case "RadioButton"

            DrawRadioButton(ctl, gr, l, t, w, h)

        Case "GroupBox"

            DrawGroupBox(ctl, gr, l, t, w, h)

        Case "ListBox"

            DrawListBox(ctl, gr, l, t, w, h)

        Case "ComboBox"

            DrawComboBox(ctl, gr, l, t, w, h)

        Case Else

            DrawUnknown(ctl, gr, l, t, w, h)

    End Select

    ' If this control contains others, draw them.

    If ctl.Controls.Count > 0 Then

        x_off += ctl.Left

        y_off += ctl.Top

        For i As Integer = ctl.Controls.Count - 1 To 0 Step -1

            DrawControl(ctl.Controls(i), gr, x_off, y_off)

        Next i

    End If

End Sub

' Draw a control that contains text and may

' have a border.

Private Sub DrawTextualControl(ByVal ctl As Control, ByVal gr As Graphics, _

 ByVal l As Integer, ByVal t As Integer, ByVal w As Integer, _

 ByVal h As Integer, ByVal align As ContentAlignment, _

 Optional ByVal force_border As Boolean = False)

    DrawBorder(ctl, gr, l, t, w, h, force_border)

    DrawText(ctl.Text, ctl.Font, align, gr, l, t, w, h)

End Sub

' Draw this control's border if it has one.

Private Sub DrawBorder(ByVal ctl As Control, ByVal gr As Graphics, ByVal l As Integer, _

 ByVal t As Integer, ByVal w As Integer, ByVal h As Integer, _

 Optional ByVal force_border As Boolean = False)

    ' Blank the control's area.

    gr.FillRectangle(Brushes.White, l, t, w, h)

    If Not force_border Then

        ' See if this type of control 

        ' has a BorderStyle property.

        Dim ctl_type As Type = ctl.GetType

        Dim borderstyle_info As System.Reflection.PropertyInfo = _

            ctl_type.GetProperty("BorderStyle")

        If Not (borderstyle_info Is Nothing) Then

            ' See if BorderStyle is None.

            force_border = (CType(borderstyle_info.GetValue(ctl, Nothing), BorderStyle) _

                <> BorderStyle.None)

        End If

    End If

    ' Draw the border.

    If force_border Then gr.DrawRectangle(Pens.Black, l, t, w, h)

End Sub

' Draw some text in this control's space

' appropriately aligned.

Private Sub DrawText(ByVal txt As String, ByVal fnt As Font, ByVal align As ContentAlignment, ByVal gr As Graphics, ByVal l As Integer, ByVal t As Integer, ByVal w As Integer, ByVal h As Integer, Optional ByVal draw_border As Boolean = False)

    Dim layout_rectangle As New RectangleF(l, t, w, h)

    Dim string_format As New StringFormat

    Select Case align

        Case ContentAlignment.BottomCenter, _

          ContentAlignment.BottomLeft, _

          ContentAlignment.BottomRight

            string_format.LineAlignment = StringAlignment.Far

        Case ContentAlignment.MiddleCenter, _

          ContentAlignment.MiddleLeft, _

          ContentAlignment.MiddleRight

            string_format.LineAlignment = StringAlignment.Center

        Case ContentAlignment.TopCenter, _

          ContentAlignment.TopLeft, _

          ContentAlignment.TopRight

            string_format.LineAlignment = StringAlignment.Near

    End Select

    Select Case align

        Case ContentAlignment.BottomCenter, _

          ContentAlignment.MiddleCenter, _

          ContentAlignment.TopCenter

            string_format.Alignment = StringAlignment.Center

        Case ContentAlignment.BottomLeft, _

          ContentAlignment.MiddleLeft, _

          ContentAlignment.TopLeft

            string_format.Alignment = StringAlignment.Near

        Case ContentAlignment.BottomRight, _

          ContentAlignment.MiddleRight, _

          ContentAlignment.TopRight

            string_format.Alignment = StringAlignment.Far

    End Select

    gr.DrawString(txt, fnt, Brushes.Black, _

        layout_rectangle, string_format)

End Sub

Listing 2. Program NetHiResPrint sends a high-resolution image of a form to a PrintPreviewDialog control.


Figure 1 shows the result. This figure demonstrates several of the program’s features including:

· Overlapping controls (e.g. on top of the PictureBoxes at the right)

· RadioButtons positioned inside GroupBoxes

· Nested GroupBoxes

· Different fonts

· Selected items in the ListBox

· Selected RadioButtons and CheckBoxes

*** NetHiResPrint Fig1.bmp ***

Figure 1. A high-resolution print preview.

Conclusion


Subroutine PrintPreviewForm makes implementing PrintForm in VB .NET a snap. With a single line of code, you can display a preview of a form’s image and let the user zoom, scroll, and print it.


Subroutine HiResPrint automatically generates a high-resolution version of a form. This produces a smoother, nicer-looking printout. It may also save time because your computer only needs to send a few drawing instructions to the printer rather than a large bitmap.


Unfortunately (at least in this context) VB .NET includes a huge number of new controls that we didn’t have in VB 6. Those controls also provide a large number of new options for formatting. For example, did you know that you can place a RadioButton’s dot in the center of the control on top of the text? Providing routines to draw all of these controls formatted in every possible way would be a huge undertaking.


Rather than trying to implement routines for every control, I’ve written routines for the controls I use most in the layouts I use most. You can write routines for other controls as you need them. Drop me an email when you do and I’ll post your improvements. If everyone pitches in, these programs can grow into a very powerful tool for easily printing VB .NET forms.

Rod’s book Visual Basic Graphics Programming, Second Edition explains how to provide similar high-resolution printing for VB 6. Get more information at www.vb-helper.com/vbgp.htm. Rod has written more than a dozen books and hundreds of magazine articles. Learn more, sign up for his newsletter, or download some of his thousands of example programs at www.vb-helper.com. RodStephens@vb-helper.com.

